program define incent, rclass
   version 15.1
	syntax varlist [if] [pweight], [OPTS(str) NODRAW]
	marksample touse	 
	if "`weight'" ~= "" {
	   local wt [`weight'`exp']
   }	
	tempname fx
	* NN = never shame
	* YY = shame 5% and 25%
	* YN = shame 5% Only
	reg `varlist' i.axn5#shamed_1v0 if `touse' & axn5~=0 `wt', robust
	margins {axn5 -.5 -.5 .5 .5}, over(shamed_1v0) // NN, YY
	mat `fx' = r(table)
	reg `varlist' i.axn5 if `touse' & ((shamed==1 & inlist(axn5,1,2)) | ///
	   (shamed==0 & inlist(axn5,3,4))) `wt', robust
   margins i.axn5 // just to see the levels		
	margins {axn5 -.5 -.5 .5 .5} // YN
	mat `fx' = `fx', r(table)
	mat colnames `fx' = nn yy yn
	if "`nodraw'" == "" {
	   local gopts scheme(s1mono) plotregion(style(none)) mlabel ///
	      format(%3.0f) mlabpos(12) legend(off) 			
	   coefplot (matrix(`fx'[1,]), ci("`fx'[5,] `fx'[6,]") msymbol(O)),  ///
		   /* order(nn yy yn) */ keep(nn yn) order(nn yn) ///
		   /* coeflabel(nn="No Shaming" yy="Blanket Shaming" yn="Selective Shaming") */ ///
			coeflabel(nn="No Shaming" yn="Shaming") ///
		   `gopts' `opts'
	}
	mat list `fx'
	return matrix fx = `fx'
end


